

Laurea Specialistica in Ingegneria Informatica

Laurea Specialistica in Ingegneria Informatica per la Gestione d'Azienda

Corso di Sistemi Intelligenti – a.a. 2005-06

Ing. Mario G.C.A. Cimino

Introduzione a Expert System Shell

eXpertise2Go.com
Web-Enabled Expert Systems (e2gLite v1.0)

Pisa, Aprile 2005

Introduzione a Expert System Shell (<http://expertise2go.com/webesie/e2gdoc/>)

a) Installazione e primo avvio

- ✓ File necessari: l'interfaccia (*interface.html*), il motore inferenziale (*engine.jar*), la base di conoscenza (*kbase.kb*) ed (opzionale) il file definizione della sintassi (*kbase.syn*) per *TextPad*©
- ✓ Installare la Java Virtual Machine (*J2SE JRE*, <http://java.sun.com/j2se/1.4.2/download.html>) ed eventualmente modificare le opzioni del web browser abilitando l'esecuzione delle applet Java.
- ✓ Per avviare l'applicazione, aprire il file *interface.html* con il web browser.
- ✓ In alternativa a *Notepad*, come programma di editing della base di conoscenza, installare *Textpad*© (<http://www.textpad.com/download/>), editor in grado di numerare le righe, colorare opportunamente le parole chiave, eseguire operazioni di sostituzione. Quindi effettuare i passi seguenti.
- ✓ Copiare il file di definizione della sintassi in *C:\Program Files\TextPad 4\system*
- ✓ Avviare *Textpad*, selezionare il menu *configure*, e poi *new document class*, selezionando i seguenti valori: *Document class name: kbase*, *Class members: *.kb*, *Enable syntax highlighting*, *Syntax definition file: kbase.kb*
- ✓ Per evitare che *Textpad* inserisca dei caratteri speciali nel testo (non permessi dall'interprete), selezionare il menu *configure*, quindi *preferences*, poi *document classes*, ed infine disabilitare le prime tre voci in elenco.

b) Esercizio 1

Creare il file di testo *kbase.kb*, dal contenuto specificato in Fig.1, ed eseguire Expert System Shell.

Osservazioni:

R1, R2, testo che identifica la regola

A1, A2, C1, C2, testo che identifica un attributo dell'antecedente o del conseguente
"V1", valore stringa

10, valore numerico

true, valore booleano

PROMPT, richiesta del valore di un attributo necessario al motore inferenziale

DEFAULT, valore assunto per l'attributo nel caso in cui il motore inferenziale non ottenga alcun valore dall'utente o dalle regole

GOAL, attributo per il quale il motore inferenziale deve identificare un valore

Domande:

Esaminare, tramite la finestra di debug, i passaggi del motore inferenziale e scoprire cosa accade nei seguenti casi:

- *A1="V1", A2=5*
- *A1=<nessuna risposta>*
- *A1="V3", A2= -50*

```
REM commento
REM -- definizione delle regole di produzione
RULE [R1]
  If [A1] = "V1" and
     [A2] < 10
  Then [C1] = true

RULE [R2]
  If [C1] = true
  Then [C2] = "Test riuscito"

REM -- definizione di richieste di fatti
PROMPT [A1] MultChoice
"I possibili valori di A1 sono:"
"V1"
"V2"
"V3"

PROMPT [A2] Numeric
"Inserire un valore tra 0 e 100 per A2"
"0"
"100"
DEFAULT [A2] = 50
DEFAULT [C1] = false

REM -- definizione degli obiettivi
GOAL [C2]
```

Fig.1

c) Esercizio 2

Creare un nuovo file di testo *kbase.kb* ed implementare la base di conoscenza relativa all'esempio di concatenazione all'indietro visto a lezione, di seguito riportato. Seguire i passi del motore inferenziale ed individuare eventuali differenze rispetto al comportamento ipotizzato.

```
Esempio
REGOLA 1      IF la pressione sanguigna è alta
                THEN il rischio di problemi cardiaci è alto
REGOLA 2      IF la pressione sanguigna è bassa
                THEN il rischio di problemi cardiaci è basso
REGOLA 3      IF l'assunzione di alcool è alta
                AND l'assunzione di sale è alta
                THEN la pressione sanguigna è alta
REGOLA 4      IF l'assunzione di alcool è bassa
                AND l'assunzione di sale è bassa
                THEN la pressione sanguigna è bassa
REGOLA 5      IF le unità di alcool per settimana sono > 40
                THEN l'assunzione di alcool è alta
REGOLA 6      IF le unità di alcool per settimana sono < 30
                THEN l'assunzione di alcool è bassa
REGOLA 7      IF le unità di alcool per settimana sono  $\epsilon$  30 e  $\delta$  40
                THEN l'assunzione di alcool è media
QUESTION      unità di alcool
QUESTION      assunzione di sale
FIND          rischio di problemi cardiaci
```

Supponiamo di voler calcolare il rischio di problemi cardiaci per un certo paziente usando la concatenazione all'indietro.

Il motore inferenziale cerca la prima regola nella base di conoscenza con "rischio di problemi cardiaci" nella sua conclusione. Trova la regola 1. Per dimostrare la conclusione della regola 1, il motore inferenziale deve provare che la condizione di tale regola è vera, cioè "la pressione sanguigna è alta". Questa condizione può essere dimostrata cercando di dimostrare la regola 3. Questo richiede di valutare le condizioni della regola 3. Per dimostrare "l'assunzione di alcool è alta", deve essere vera la condizione nella regola 5, cioè "le unità di alcool per settimana sono > 40". Tale condizione può essere valutata ponendo una domanda all'utente. Se l'utente immette un valore > 40, la regola 5 ha successo. La prossima condizione nella regola 3, cioè "l'assunzione di sale è alta", non è conclusione di alcuna regola, quindi si pone un'altra domanda all'utente. Se la risposta data dall'utente è "alta", la regola 3 ha successo e conseguentemente anche la regola 1. Se invece, la risposta dell'utente è "bassa", la regola 3 fallisce. In questo caso, il motore inferenziale cerca la prossima regola nella base di conoscenza con "rischio di problemi cardiaci" nella conclusione. Trova la regola 2. Il motore inferenziale cerca di provare la regola 2 così come aveva fatto con la regola 1.

d) Altri Esercizi

Esaminare e provare gli esempi *auto.kb*, *autofr.kb*, *admit.kb*, *wine.kb*. Ricercare nella cartella *userguide* il significato di ulteriori costrutti linguistici riscontrati, e discuterne eventualmente con il docente.

E' possibile indicare al motore inferenziale un nome differente per la base di conoscenza. Selezionando dal browser il menu *view* e poi *source*, appare una finestra con il codice html, dove e' facile identificare un elenco di attributi quali: il motore inferenziale, le dimensioni della finestra, il nome della base di conoscenza, titolo e sottotitolo della finestra, colore di sfondo e del testo (valori RGB esadecimali), finestra di debug. Effettuare le opportune modifiche, salvare il file e cliccare sul tasto di *reload* del browser per riavviare il sistema.

e) Soluzione Esercizio 2

```
REM kbase.kb

REM valutazione del rischio -----
RULE [R1 Il rischio di problemi cardiaci e` alto?]
  If [la pressione sanguigna] = "alta"
  Then [il rischio di problemi cardiaci] = "alto"

RULE [R2 Il rischio di problemi cardiaci e` basso?]
  If [la pressione sanguigna] = "bassa"
  Then [il rischio di problemi cardiaci] = "basso"

REM deduzioni sulla pressione sanguigna -----
RULE [R3 La pressione sanguigna e` alta?]
  If [l'assunzione di alcool] = "alta" and
     [l'assunzione di sale] = "alta"
  Then [la pressione sanguigna] = "alta"

RULE [R4 La pressione sanguigna e` bassa?]
  If [l'assunzione di alcool] = "bassa" and
     [l'assunzione di sale] = "bassa"
  Then [la pressione sanguigna] = "bassa"

REM deduzioni sulla assunzione di sostanze -----
RULE [R5 L'assunzione di alcool e` alta?]
  If [le unita` di alcool per settimana] > 40
  Then [l'assunzione di alcool] = "alta"

RULE [R6 L'assunzione di alcool e` bassa?]
  If [le unita` di alcool per settimana] < 30
  Then [l'assunzione di alcool] = "bassa"

RULE [R7 L'assunzione di alcool e` media?]
  If [le unita` di alcool per settimana] > 29 and
     [le unita` di alcool per settimana] < 41
  Then [l'assunzione di alcool] = "media"

REM richiesta di fatti -----
PROMPT [le unita` di alcool per settimana] Numeric
"Quante unita` di alcool per settimana vengono assunte? (0..100)"
"0"
"100"

PROMPT [l'assunzione di sale] MultChoice
"L'assunzione di sale e:"
"bassa"
"alta"

REM obiettivi -----
GOAL [il rischio di problemi cardiaci]
-----
```

NOTA: il costrutto **DEFAULT** viene usato anche per esprimere fatti. Esempio: **DEFAULT [A2] = 5.**